
Abstract
This paper derives an abstract machine and a virtual machine for the λ-calculus with four variants of
delimited-control operators: shift/reset, control/prompt, shift0/reset0, and control0/prompt0.
Starting from Shan’s definitional interpreter for the four operators, we successively apply various
meaning-preserving transformations. Both trails of invocation contexts (needed for control and
control0) and metacontinuations (needed for shift0 and control0) are defunctionalized and
eventually represented as a list of stack frames. The resulting virtual machine clearly models not
only how the control operators and captured continuations behave but also when and which portion
of stack frames is copied to the heap.

1 Introduction

Manipulation of control structure of a program is inevitable. In addition to the standard
exception handling, more sophisticated manipulation of control using algebraic effects and
handlers has been proposed [4, 25] and is becoming widely used [20]. To support such
mechanisms in a compiler, one can either (i) transform the source program into continuation-
passing style (CPS), or (ii) implement manipulation of control directly via the modification of
a portion of a stack without transforming the program into CPS. There is extensive research
comparing which approach (among more variants) is better in which circumstances [12].

However, for four variants of delimited-control operators, i.e., shift and reset [8, 9],
control and prompt [13], shift0 and reset0 [23], and control0 and prompt0 [16], almost no
low-level implementation has been considered. The only exceptions we are aware of are all on
shift/reset: direct implementation of shift/reset in Scheme48 [15], in OchaCaml [22],
and the derivation of a virtual machine for shift/reset [3]. Without proper low-level
implementation strategies for all the four delimited-control operators, we cannot even discuss
pros and cons of CPS vs. direct-style implementations for those operators. This omission
could affect the low-level implementation strategies for algebraic effects and handlers, since
they have a close connection with shift0 and control0 [14, 24].

16:2 Virtual Machine for Four Delimited-Control Operators

In this paper, we derive an abstract machine and a virtual machine for the λ-calculus
with four delimited-control operators. Starting from Shan’s definitional interpreter [28], we
successively apply various meaning-preserving transformations, following Danvy’s recipe [2, 7].
The overall derivation is similar to our previous work [3] on deriving a virtual machine for
shift/reset. However, handling of invocation contexts (needed for control and control0)
and metacontinuations (needed for shift0 and control0) is non-trivial: we need to have
a trail of invocation contexts to be a tree structure to support concatenation of invocation
contexts and have a metacontinuation to maintain a list of code pointers representing the
contexts outside delimiters.

In summary, we make the following contributions in this paper:

We present the first virtual machine that supports four delimited-control operators and
that explains how they manipulate stacks.

We show it is possible to apply Danvy’s method of inter-deriving semantic artifacts to
four delimited-control operators, giving another non-trivial example and widening its
applicability.

We clarify how trails and metacontinuations can be represented in a stack, suggesting a
low-level implementation strategy for four delimited-control operators.

After introducing four delimited-control operators in the next section, we first show the
definitional interpreter in Section 3. We then apply various program transformation to obtain
a stack-based interpreter in Section 4, showing an abstract machine in passing. In Sections 5
and 6, we derive a compiler and a virtual machine. Related work is discussed in Section 7
and the paper concludes in Section 8. The appendix shows an example how a program is
compiled to a list of instructions and executed on the virtual machine. The omitted OCaml
code is available as supplementary material.

2 Four Delimited-Control Operators

Delimited-control operators enable us to capture the current continuation up to the enclosing
delimiter and use it in the subsequent program. There are four variants of delimited-control
operators: shift (S) and reset [8, 9], control (F) and prompt [13], shift0 (S0) and reset0
[23], and control0 (F0) and prompt0 [16]. Since the behavior of all the four delimiters
(reset, prompt, reset0, and control0) are exactly the same, we use a uniform notation ⟨⟩
for them. The basic behavior of the four operators are to capture the current continuation up
to the enclosing delimiter and execute their body. We describe their exact behavior below.

A shift expression, Sc. e, clears the current continuation up to the enclosing delimiter,
binds it to c, and execute e. Thus, in 1 + ⟨(Sc. 2 × c 3) + 4⟩, the continuation ⟨[] + 4⟩ is
cleared, bound to c, and 2 × c 3 is executed in reset. The original expression reduces to
1 + ⟨2 × c 3⟩, giving the final result 15.

A control expression, Fc. e, differs from shift in that it does not insert a delimiter
into the captured continuation. In 1 + ⟨(Fc. 2 × c 3) + 4⟩, c is bound to [] + 4 without
surrounding reset. If the captured continuation contains another control, as in 1 +
⟨(Fc. 2 × c 3) + Fc′. 4⟩, c is bound to [] + Fc′. 4. The original expression reduces to 1 +
⟨2 × (3 + Fc′. 4)⟩, where the second F captures (and discards) not just 3 + [] but also the
invocation context of c, namely 2 × [], giving the final result 5. Using F , one can access the
context in which the captured continuation is invoked. This is in contrast to the shift case:
1 + ⟨(Sc. 2 × c 3) + Sc′. 4⟩ reduces to 1 + ⟨2 × ⟨3 + Sc′. 4⟩⟩, giving the final result 9. Using

more than one F in the same context, we can capture multiple invocation contexts.1 To
account for the invocation contexts of captured continuations, an interpreter for F must
maintain a trail of continuations [5].

A shift0 expression, S0c. e, on the other hand, removes the original reset surrounding
the shift0 expression (but retains the reset around the captured continuation as in S).
By nesting S0, one can access the context outside the enclosing reset. For example,
⟨1 + ⟨(S0c. S0c′. 2 × c′3) + 4⟩⟩ reduces to ⟨1 + (S0c′. 2 × c′3)⟩ where c is bound to ⟨[] + 4⟩ but
is discarded. Note that there is no reset around S0c′. 2 × c′3. Thus, c′ is bound to the
context ⟨1 + []⟩, which was outside the original reset, giving the final result 8. This is
in contrast to the shift case: ⟨1 + ⟨(Sc. Sc′. 2 × c′3) + 4⟩⟩ reduces to ⟨1 + ⟨(Sc′. 2 × c′3)⟩⟩.
Now, c′ is bound to an empty context [], giving the final result 7. With more nested
occurrences of S0, arbitrarily outer contexts can be captured. To account for hierarchical
contexts, the interpreter for S0 must maintain a metacontinuation [23].

A control0 expression, F0c. e, has both the characteristics of F and S0: the captured con-
tinuation does not come with a surrounding reset and the original reset is removed. As such,
the interpreter for F0 must maintain both a trail of continuations and a metacontinuation.

Shan [28] provides a detailed explanation on the difference between the four control
operators, as well as an example where the choice of the four operators results in four
different result values. Dyvbig, Peyton Jones, and Sabry [11] explain the four delimited-
control operators in terms of different primitive control operators.

3 The Definitional Interpreter

Listing 1 shows the definitional interpreter for the λ-calculus extended with four delimited-
control operators and the delimiter, written in OCaml. The interpreter is written in
continuation-, trail-, and metacontinuation-passing style. Although the main interpreter
function f1 receives a trail and a metacontinuation explicitly, they do not play any roles for
the pure λ-calculus terms. If we η-reduce them, the definition coincides with the standard
continuation-passing style interpreter.

As in our previous work [3], an environment is represented as two lists, a list of variable
names xs and a list of values vs, instead of an association list. This design comes from the
goal of this work. Since we will decompose the interpreter into a compiler and a virtual
machine, we separate an environment into the part that depends only on the input term and
the part that depends on runtime values. The function Env.offset returns the offset of a
variable within a given list.

In the interpreter, the current continuation and trail in the innermost surrounding
delimiter are stored in the arguments c and t (of types c and t, respectively), while the
continuations and trails outside the delimiter are stored in metacontinuation m, which is a
list2 of pairs of a continuation and a trail of each context. Thus, the context is delimited
(in the Reset (e) case) by storing c and t to m and evaluating the body e in the initial
continuation idc and the empty trail TNil.

To capture the current continuation and trail, one of four control operators is used. In
all four cases, the current continuation c and trail t are captured, bound to x, and the body
of the control operator is evaluated under appropriate settings.

1 See [6] for the general case as well as other (typed) examples of the use of F .
2 We use MNil and MCons to construct metacontinuations. We cannot use (c * t) list as the definition

of m, because the types c and m would then be circular.

16:4 Virtual Machine for Four Delimited-Control Operators

Listing 1 The definitional interpreter.
(* syntax *)
type e = Var of string | Fun of string * e | App of e * e

| Shift of string * e | Control of string * e
| Shift0 of string * e | Control0 of string * e
| Reset of e

type v = VFun of (v -> c -> t -> m -> v) (* value *)
| VContS of c * t | VContC of c * t

and c = v -> t -> m -> v (* continuation *)
and t = TNil | Trail of (v -> t -> m -> v) (* trail *)
and m = MNil | MCons of (c * t) * m (* metacontinuation *)

(* initial continuation : v -> t -> m -> v *)
let idc v t m = match t with

TNil -> (match m with
MNil -> v

| MCons ((c,t),m) -> c v t m)
| Trail(h) -> h v TNil m

(* cons : (v -> t -> m -> v) -> t -> t *)
let rec cons h t = match t with

TNil -> Trail(h)
| Trail(h’) -> Trail(fun v t’ m -> h v (cons h’ t’) m)

(* apnd : t -> t -> t *)
let apnd t0 t1 = match t0 with

TNil -> t1
| Trail(h) -> cons h t1

(* f1 : e -> string list -> v list -> c -> t -> m -> v *)
let rec f1 e xs vs c t m = match e with

Var(x) -> c (List.nth vs (Env. offset x xs)) t m
| Fun(x,e) ->

c (VFun(fun v c’ t’ m’ -> f1 e (x::xs) (v::vs) c’ t’ m ’)) t m
| App(e0 ,e1) ->

f1 e0 xs vs (fun v0 t0 m0 ->
f1 e1 xs vs (fun v1 t1 m1 ->
(match v0 with

VFun(f) -> f v1 c t1 m1
| VContS (c’,t’) -> c’ v1 t’ (MCons ((c,t1),m1))
| VContC (c’,t’) -> c’ v1 (apnd t’ (cons c t1)) m1)) t0 m0) t m

| Shift(x,e) -> f1 e (x::xs) ( VContS (c,t):: vs) idc TNil m
| Control (x,e) -> f1 e (x::xs) ( VContC (c,t):: vs) idc TNil m
| Shift0 (x,e) -> (match m with

MCons ((c0 ,t0),m0) -> f1 e (x::xs) ( VContS (c,t):: vs) c0 t0 m0)
| Control0 (x, e) -> (match m with

MCons ((c0 ,t0),m0) -> f1 e (x::xs) ( VContC (c,t):: vs) c0 t0 m0)
| Reset(e) -> f1 e xs vs idc TNil (MCons ((c,t),m))

(* f : e -> v *)
let f expr = f1 expr [] [] idc TNil MNil

For Shift (x, e) and Control (x, e), the body e is evaluated under the initial con-
tinuation and the empty trail. This reflects the fact that the original reset surrounding
the control operator remains for these cases. Even if we use control operators within e,
we cannot access the contexts outside reset because they reside in m.
For Shift0 (x, e) and Control0 (x, e), on the other hand, the body e is evaluated
under the topmost continuation and trail stored in the metacontinuation m.3 This reflects
the fact that the original reset surrounding the control operator is removed for these cases.
By using control operators within e, we can access the context outside the innermost
reset.

The captured continuation and trail are packaged into VContS for Shift (x, e) and
Shift0 (x, e) and into VContC for Control (x, e) and Control0 (x, e). When VContS
or VContC is applied (in the App case), it behaves differently depending on whether reset is
present around the invocation.

For VContS (c’, t’), the continuation c and trail t1 at the invocation time are pushed
into metacontinuation m1. This reflects the fact that the invocation of a continuation
captured by Shift (x, e) or Shift0 (x, e) is surrounded by reset. Even if we use
control operators within c’, we cannot access c and t1 because they reside in the
metacontinuation.
For VContC (c’, t’), on the other hand, the continuation c and trail t1 at the invocation
time are concatenated to the current trail t’. This reflects the fact that the invocation
of a continuation captured by Control (x, e) or Control0 (x, e) is not surrounded
by reset; since the invocation-time continuation and trail are put into the trail, they
can be captured by using control operators within c’.

Adding a continuation to a trail and appending two trails are realized by cons and
apnd, respectively. A trail is either an empty trail TNil or a non-empty trail Trail holding
a continuation, which represents functional composition of all the invocation contexts
(continuations) encountered so far.

The interpreter is identical to Shan’s interpreter [28] except for two points. First, Shan
uses higher-order functions directly to represent captured continuations, while we use a
defunctionalized form. We could have started from the higher-order functions; by applying
defunctionalization to it, we obtain Listing 1. Second, Shan concatenates the captured
continuation c’ and trail t’ with the continuation c and trail t1 at the invocation time as
((cons c’ t’) v1 (cons c t1)). By case analysis on t’, it is straightforward to verify
that Shan’s code is equivalent to (c’ v1 (apnd t’ (cons c t1))) which we adopt. The
latter is also used by Biernacki, Danvy, and Millikin [5] and Kameyama and Yonezawa [19].

4 Stack Introduction

In this and next sections, we successively apply meaning-preserving program transformations
to the definitional interpreter to obtain a compiler and a virtual machine. In this section, we
introduce a stack into the interpreter by (1) defunctionalizing continuations (Section 4.1),
(2) linearizing them into a list of frames (Section 4.2), and (3) separating static and dynamic
data in the frames (Section 4.3). Along the way, we derive a stack-based abstract machine
(Section 4.5).

3 Metacontinuation m must be non-empty here. Otherwise, a pattern-match error is raised. (In the
supplementary material, a more sensible error message “shift0/control0 is used without enclosing reset”
is given.)

16:6 Virtual Machine for Four Delimited-Control Operators

Listing 2 Type definition for defunctionalized interpreter.
type v = VFun of (v -> c -> t -> m -> v)

| VContS of c * t | VContC of c * t
and c = C0 | CApp0 of e * string list * v list * c | CApp1 of v * c
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * t) * m

Listing 3 Type definition for linearized interpreter.
type v = VFun of (v -> c -> t -> m -> v)

| VContS of c * t | VContC of c * t
and f = CApp0 of e * string list * v list | CApp1 of v
and c = f list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * t) * m

4.1 Defunctionalization
We first defunctionalize [26, 27] continuations in the definitional interpreter. In Listing 1,

the type c is higher order. We turn it into a datatype as shown in Listing 2. The identity
continuation is represented as C0, while two continuations in the App case are represented as
App0 and App1 where the arguments represent free variables of the respective continuations.
The resulting datatype essentially represents evaluation contexts.

We do not defunctionalize the argument of VFun at this point, because it is not necessary
for stack introduction. This choice is arbitrary: we could defunctionalize it and the rest of
derivations would go through without any problem. We will defunctionalize it later when we
need to do so, to derive an abstract machine and a virtual machine.

We do not defunctionalize the argument of Trail, either. Even though the type of
the argument of Trail is the same as c, defunctionalizing it together with c leads to tree-
structured continuations. We can still obtain the same abstract machine and virtual machine,
but by defunctionalizing it separately at a later stage, we can keep the definition of c to
have a list structure (as in our previous work [3]) and postpone the introduction of a tree
structure until Section 5.3.

We omit the standard definition of the defunctionalized interpreter due to the lack of
space; see the supplementary material. We simply introduce a dispatch function for c and use
it whenever a continuation is applied. The transformation is the standard defunctionalization
and thus the resulting interpreter behaves the same as the definitional interpreter.

4.2 Linearizing Continuations
The type c in Listing 2 is isomorphic to a list where C0 is an empty list and CApp0 and

CApp1 are conses. Thus, we linearize continuations, i.e., we transform c into an OCaml list
as shown in Listing 3. The type c is now a list of frames, where a frame f stores data that
were previously held in CApp0 and CApp1.

Obviously, the new interpreter (omitted) behaves the same as the previous one.

4.3 Introducing Stacks
Examining the type f in Listing 3, we notice that the constructors CApp0 and CApp1

contain both static (compile-time) and dynamic (run-time) data. Static data include the
term e and the variable list string list in CApp0, which are fixed once the input program

Listing 4 Type definition for stack-based interpreter.
type v = VFun of (v -> c -> s -> t -> m -> v)

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list

and f = CApp0 of e * string list | CApp1
and c = f list
and s = v list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * s * t) * m

Listing 5 Type definition for delinearized interpreter.
type v = VFun of (v -> c -> s -> t -> m -> v)

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list

and c = C0 | CApp0 of e * string list * c | CApp1 of c
and s = v list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * s * t) * m

is given. Dynamic data include v list in CApp0 and v in CApp1, which are available only
at run-time. Since our goal is to transform the interpreter into a compiler and a virtual
machine, we separate these two types of data by introducing a stack.

Listing 4 shows the resulting data definition. The previous continuation c is split into
a pair of a continuation c and a stack s. The former is a list of frames, where the frame
f now keeps only the static data. The runtime data are kept in the stack, which is a list
of values. Since the previous CApp0 included v list, the value v is extended with VEnv to
store the v list as a single value.4 Since the new c (a list of frames) and s (a list of values)
are obtained by splitting a single list (a list of f in Listing 3), they always have the same
length. In the subsequent derivations, we keep this invariant throughout.

Because we only changed the representation of c locally, we immediately see that the
new interpreter behaves the same as the previous one.

4.4 Delinearizing Continuations
The purpose of defunctionalization (Section 4.1) and linearization of continuations (Sec-

tion 4.2) was to introduce a data stack. Now that we have introduced a data stack, we
transform continuations back to the higher-order form via delinearization. In this section,
we convert lists into constructors.

Listing 5 shows the resulting data definition. Here, only the static f is incorporated into
c. The stack s remains as a list of values. Note that c contains only static data (in contrast
to c in Listing 2 that contains both static and dynamic data). All the dynamic data are still
carried around in s. As in Section 4.2, the old and new representations of c are isomorphic,
and thus the new interpreter behaves the same as the previous one.

4 The introduction of VEnv into v is arbitrary. Although we introduced it to emulate caller-save registers
often found in the compiled code, a user cannot write a program that evaluates to VEnv. Instead, we
could introduce a new type for stack items that consists of either a value or a list of values (VEnv). In
the current paper, we followed our previous work [3] and included VEnv directly to v.

16:8 Virtual Machine for Four Delimited-Control Operators

Figure 1 Abstract machine.

e ⇒ ⟨e, [], [], C0, [], TNil, []⟩
⟨x, xs, vs, c, s, t, m⟩ ⇒ ⟨c, List.nth vs (offset x xs), s, t, m⟩

⟨λx.e, xs, vs, c, s, t, m⟩ ⇒ ⟨c, VFun(e, x, xs, vs), s, t, m⟩
⟨e0 e1, xs, vs, c, s, t, m⟩ ⇒ ⟨e0, xs, vs, CApp0(e1, xs, c), VEnv(vs) :: s, t, m⟩

⟨Shift(x, e), xs, vs, c, s, t, m⟩ ⇒ ⟨e, x :: xs, VContS(c, s, t) :: vs, C0, [], TNil, m⟩
⟨Control(x, e), xs, vs, c, s, t, m⟩ ⇒ ⟨e, x :: xs, VContC (c, s, t) :: vs, C0, [], TNil, m⟩

⟨Shift0(x, e), xs, vs, c, s, t, (c0, s0, t0) :: m0⟩ ⇒ ⟨e, x :: xs, VContS(c, s, t) :: vs, c0, s0, t0, m0⟩
⟨Control0(x, e), xs, vs, c, s, t, (c0, s0, t0) :: m0⟩ ⇒ ⟨e, x :: xs, VContC (c, s, t) :: vs, c0, s0, t0, m0⟩

⟨Reset(e), xs, vs, c, s, t, m⟩ ⇒ ⟨e, xs, vs, C0, [], TNil, (c, s, t) :: m⟩
⟨C0, v, [], TNil, []⟩ ⇒ v

⟨C0, v, [], TNil, (c, s, t) :: m⟩ ⇒ ⟨c, v, s, t, m⟩
⟨C0, v, [], Trail(h), m⟩ ⇒ ⟨h, v, TNil, m⟩

⟨CApp0(e, xs, c), v, VEnv(vs) :: s, t, m⟩ ⇒ ⟨e, xs, vs, CApp1(c), v :: s, t, m⟩
⟨CApp1 (c), v, VFun(e, x, xs, vs) :: s, t, m⟩ ⇒ ⟨e, x :: xs, v :: vs, c, s, t, m⟩

⟨CApp1(c), v, VContS(c′, s′, t′) :: s, t, m⟩ ⇒ ⟨c′, v, s′, t′, (c, s, t) :: m⟩
⟨CApp1(c), v, VContC (c′, s′, t′) :: s, t, m⟩ ⇒ ⟨c′, v, s′, apnd t′ (cons (Hold (c, s)) t), m⟩

⟨Hold (c, s), v, t, m⟩ ⇒ ⟨c, v, s, t, m⟩
⟨Append (h, h′), v, t, m⟩ ⇒ ⟨h, v, cons h′ t, m⟩

4.5 Abstract Machine

In this section, we briefly describe the abstract machine that can be derived from the
interpreter in Section 4.4. Since all the interpreters in this paper receive a continuation
and a metacontinuation, all the calls to interpreter functions (such as f1 and the dispatch
function for continuations) are tail calls. As such, we can easily derive an abstract machine
by simply regarding the arguments to interpreter functions as a state of the abstract machine.
The derived abstract machine is shown in Figure 1. Although we omit the code for the
interpreter, one can imagine how it looks like from the abstract machine. To extract the
abstract machine, we further performed the following transformations:

We defunctionalized the argument of VFun. A function is now represented as a closure.
We will perform the same transformation later; see Section 5.3.
We defunctionalized the argument of Trail. The Trail data are constructed in the
two branches of cons (see Listing 1). The first one is represented as Hold that holds an
invocation context; the second one as Append that appends two trails. We will perform
the same transformation later; see Section 5.3 for details.
Instead of MNil and MCons, we use standard lists for metacontinuations.

Because we have introduced a stack into the interpreter, we obtain a stack-based abstract
machine. This is in contrast to the previous abstract machines [5, 11, 28] which do not carry
a stack explicitly. The obtained abstract machine clearly describes the behavior of control
operators. When one of the control operators is used, the current continuation c, stack s,
and trail t are captured, put into a stack, and bound to x. Then, the body of the control
operator is executed. For Shift and Control, the current continuation and trail are cleared,
whereas for Shift0 and Control0, the ones in the metacontinuation are used. The reset
operator pushes the current c, s, and t on the metacontinuation m, and initializes them.

When a continuation captured by Shift or Shift0 is invoked, the current c, s, and t are
pushed onto m and the captured state is reinstated. When a continuation captured by
Control or Control0 is invoked, on the other hand, t is extended by c and s (via cons), and
the result is in turn extended by t′ (via apnd).

4.6 Refunctionalizing Continuations
Finally, Listing 6 shows the refunctionalized interpreter where defunctionalized con-

tinuations are transformed back to higher-order functions. It is similar to the definitional
interpreter in Listing 1, but passes around a stack. Typewise, all the occurrences of a continu-
ation c in Listing 1 are replaced by pairs c * s of a continuation and a stack. Furthermore,
the type c and the type of the argument of VFun are modified to receive a stack.

Compared to the definitional interpreter f1 in Listing 1, the refunctionalized interpreter
f6 receives an additional stack argument s, and whenever it returns a value, a continuation
c is applied to the value together with a stack s. We can also observe that the references
to free variables in the definitional interpreter (vs and v0 in the App case) are now realized
by passing those values via the stack. We push those values at the recursive calls and pop
them when the corresponding continuations are called. Since stacks are extracted from
continuations and stacks have the same structure as (now refunctionalized) continuations,
popping a value would never fail: popped values correspond to the dynamic arguments of
CApp0 and CApp1. This is the consequence of the invariant we keep between continuations
and stacks. Similarly, idc corresponds to C0, which has no dynamic counterpart. Thus, the
stack argument of idc (the second argument of idc in Listing 6) must be an empty stack.

The argument of Trail needs explanation. Since we have not defunctionalized the
argument of Trail yet, we need type conversion to store a continuation c in a trail. See the
first argument to cons in the App case. The continuation c is turned into fun v t m -> c
v s1 t m with s1 being a free variable. Later when we defunctionalize it, the stack s1 will
be extracted; see Section 5.3.

It is not straightforward to obtain the refunctionalized interpreter from the previous one.
One has to verify that the previous interpreter is in defunctionalized form [10]. However, once
it is obtained, it is simple to verify its correctness: by defunctionalizing the refunctionalized
interpreter, we can obtain the previous one.

5 Deriving a Virtual Machine

In this section, we derive a virtual machine from the refunctionalized interpreter obtained in
Section 4.6. We first combine arguments so that values are passed via a stack (Section 5.1).
We then stage the interpreter into a compiler that operates on instructions represented as
functions (Section 5.2). By defunctionalizing the instructions (Section 5.3) and linearizing
instructions (Section 5.4) and stacks (Section 5.5), we obtain the virtual machine (Section 6).

5.1 Combining Arguments
In Listing 6, functions in VFun as well as continuations c receive both a value v and a stack
s. In a low-level implementation, such as a virtual machine, we want to pass all the values
via a stack rather than passing a value and a stack separately. Listing 7 shows the type
definition of the result of such a transformation.

The argument v is removed from the argument of VFun and c. When we call such a
function, we push v to the stack before the call. When the function is called, we pop v from
the stack before the function body is executed. We do the same for the interpreter function:
we remove the vs argument and push it on the stack. As a result, the type of the interpreter
function f7 after the transformation becomes as follows:

(* f7 : e -> string list -> c -> s -> t -> m -> v *)

Listing 6 Refunctionalized interpreter (cons and apnd are the same as in Listing 1).
type v = VFun of (v -> c -> s -> t -> m -> v)

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list

and c = v -> s -> t -> m -> v
and s = v list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * s * t) * m

(* initial continuation : v -> s -> t -> m -> v *)
let idc v [] t m = match t with

TNil -> (match m with
MNil -> v

| MCons ((c,s,t),m) -> c v s t m)
| Trail(h) -> h v TNil m

(* f6 : e -> string list -> v list -> c -> s -> t -> m -> v *)
let rec f6 e xs vs c s t m = match e with

| Var(x) -> c (List.nth vs (Env. offset x xs)) s t m
| Fun(x,e) ->

c (VFun(fun v c’ s’ t’ m’ -> f6 e (x::xs) (v::vs) c’ s’ t’ m ’))
s t m

| App(e0 ,e1) ->
f6 e0 xs vs (fun v0 (VEnv(vs ):: s0) t0 m0 ->
f6 e1 xs vs (fun v1 (v0::s1) t1 m1 ->
(match v0 with

VFun(f) -> f v1 c s1 t1 m1
| VContS (c’,s’,t’) -> c’ v1 s’ t’ (MCons ((c,s1,t1),m1))
| VContC (c’,s’,t’) ->

c’ v1 s’ (apnd t’ (cons (fun v t m -> c v s1 t m) t1)) m1))
(v0::s0) t0 m0) (VEnv(vs )::s) t m

| Shift(x,e) -> f6 e (x::xs) ( VContS (c,s,t):: vs) idc [] TNil m
| Control (x,e) -> f6 e (x::xs) ( VContC (c,s,t):: vs) idc [] TNil m
| Shift0 (x,e) -> (match m with

MCons ((c0,s0,t0),m0) ->
f6 e (x::xs) ( VContS (c,s,t):: vs) c0 s0 t0 m0)

| Control0 (x,e) -> (match m with
MCons ((c0,s0,t0),m0) ->

f6 e (x::xs) ( VContC (c,s,t):: vs) c0 s0 t0 m0)
| Reset(e) -> f6 e xs vs idc [] TNil (MCons ((c,s,t),m))

(* f : e -> v *)
let f expr = f6 expr [] [] idc [] TNil MNil

Listing 7 Type definition for interpreter with combined arguments.
type v = VFun of (c -> s -> t -> m -> v)

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list

and c = s -> t -> m -> v
and s = v list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * s * t) * m

Since we simply changed the way two arguments are passed locally, we immediately see
that the new interpreter behaves the same as the previous one.

5.2 Introducing Combinators as Instructions
In this section, we extract a compiler from the interpreter. Looking at the type of f7

in the previous section, we notice that the first two arguments are static and the rest of
the arguments are dynamic. We first define the type i of instructions (in Listing 8) as the
dynamic part of the interpreter, which represents the work to be done when dynamic data
are received. We then regard the interpreter as a compiler that accepts two static data and
returns an instruction. Listing 8 shows the result.

The interpreter function f8, or a compiler, processes only the static data: the input term
e and a list of variable names xs. It then produces an instruction of type i, which performs
the rest of the work when dynamic data are given.

For example, in the case of Var (x), the compiler emits an instruction access, which,
given dynamic data, returns the corresponding value in the environment. In the case of App
(e0, e1), we define push_env, pop_env, and call, and concatenate these instructions using
(>>). We employ the same technique as the previous work [3]: we store the return address
VK (added to the definition of v) to the stack in return and retrieve it in call.

This interpreter behaves the same as the previous one, because if we inline all the
instructions, we obtain the interpreter in the previous section.

5.3 Defunctionalizing Instructions
In this section, we defunctionalize the functional instructions introduced in the previous

section into the ones that are closer to machine instructions. Specifically, we defunction-
alize the argument of VFun, i, c, and the argument of Trail, separately, and change the
representation of m. See Listing 9.

First, the argument of VFun (see push_closure and call in Listing 8) is defunctionalized
to a closure. Second, the instruction i is defunctionalized. All the functional instructions are
turned into constructors as shown in i in Listing 9. The corresponding dispatch function
(omitted) is a virtual machine: given an instruction and the current dynamic state, it
performs necessary operations. Observe how a virtual machine is naturally derived by
defunctionalizing functional instructions. Note also that the instruction is not linear: it
includes ISeq corresponding to (>>) and thus has a tree structure.

Third, c is defunctionalized. There are two cases that constitute the value of c in Listing 8:
the identity continuation idc, which is closed, and the second argument to i0 in (>>), fun
s’ t’ m’ -> i1 c s’ t’ m’. Since the free variables of the latter are i1 and c, we can
represent c as a list of i, regarding the former as an empty list and the latter as cons list.

Fourth, the argument of Trail is defunctionalized and given a new type h. The Trail
data are constructed in the two branches of cons (see Listing 1): its argument is either a
continuation h or fun v t’ m -> h v (cons h’ t’) m which has h and h’ as free variables.
They are represented as Hold and Append in Listing 9, respectively. Note that h has a tree
structure. Finally, the metacontinuation m is turned into an OCaml list, as no circular
dependency arises any more.5

Since all these changes are instances of defunctionalization and a simple local change of
data representation, the behavior of the new interpreter is the same as the previous one.

5 Unlike the definitional interpreter. See footnote 2.

Listing 8 Interpreter using combinators factored as instructions.
type v = VFun of (c -> s -> t -> m -> v)

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list | VK of c

and c = s -> t -> m -> v
and s = v list
and t = TNil | Trail of (v -> t -> m -> v)
and m = MNil | MCons of (c * s * t) * m
type i = c -> s -> t -> m -> v

(* (>>) : i -> i -> i *)
let (>>) i0 i1 =

fun c s t m -> i0 (fun s’ t’ m’ -> i1 c s’ t’ m’) s t m

(* instructions *)
let access n = fun c (VEnv(vs )::s) t m -> c (( List.nth vs n)::s) t m
let push_closure i = fun c (VEnv(vs )::s) t m ->

c (VFun(fun c’ (v::s’) t’ m’ -> i c’ (VEnv(v::vs )::s’) t’ m ’)::s)
t m

let return = fun _ (v::VK(c)::s) t m -> c (v::s) t m
let push_env = fun c (VEnv(vs )::s) t m ->

c (VEnv(vs ):: VEnv(vs )::s) t m
let pop_env = fun c (v:: VEnv(vs )::s) t m -> c (VEnv(vs )::v::s) t m
let call = fun c (v1::v0::s) t m -> match v0 with

VFun(f) -> f idc (v1::VK(c)::s) t m
| VContS (c’,s’,t’) -> c’ (v1::s’) t’ (MCons ((c,s,t),m))
| VContC (c’,s’,t’) ->

c’ (v1::s’) (apnd t’ (cons (fun v t m -> c (v::s) t m) t)) m
let shift i = fun c (VEnv(vs )::s) t m ->

i idc (VEnv( VContS (c,s,t):: vs )::[]) TNil m
let control i = fun c (VEnv(vs )::s) t m ->

i idc (VEnv( VContC (c,s,t):: vs )::[]) TNil m
let shift0 i = fun c (VEnv(vs )::s) t (MCons ((c0 ,s0 ,t0),m0)) ->

i c0 (VEnv( VContS (c,s,t):: vs ):: s0) t0 m0
let control0 i = fun c (VEnv(vs )::s) t (MCons ((c0 ,s0 ,t0),m0)) ->

i c0 (VEnv( VContC (c,s,t):: vs ):: s0) t0 m0
let reset i = fun c (VEnv(vs )::s) t m ->

i idc (VEnv(vs )::[]) TNil (MCons ((c,s,t),m))

(* f8 : e -> string list -> i *)
let rec f8 e xs = match e with

Var(x) -> access (Env. offset x xs)
| Fun(x,e) -> push_closure ((f8 e (x::xs)) >> return )
| App(e0 ,e1) ->

push_env >> (f8 e0 xs) >> pop_env >> (f8 e1 xs) >> call
| Shift(x,e) -> shift (f8 e (x::xs))
| Control (x,e) -> control (f8 e (x::xs))
| Shift0 (x,e) -> shift0 (f8 e (x::xs))
| Control0 (x,e) -> control0 (f8 e (x::xs))
| Reset(e) -> reset (f8 e xs)

(* f : e -> v *)
let f expr = f8 expr [] idc (VEnv ([])::[]) TNil MNil

Listing 9 Type definition for interpreter with defunctionalized instructions and continuations.
type v = VFun of i * v list

| VContS of c * s * t | VContC of c * s * t
| VEnv of v list | VK of c

and i = IAccess of int | IPush_closure of i | IReturn
| IPush_env | IPop_env | ICall
| IShift of i | IControl of i | IShift0 of i | IControl0 of i
| IReset of i | ISeq of i * i

and c = i list
and s = v list
and h = Hold of c * s | Append of h * h
and t = TNil | Trail of h
type m = (c * s * t) list

Listing 10 The function flat to remove ISeq.
(* flat: i -> i list *)
let rec flat i = match i with

IAccess (n) -> [ IAccess (n)]
| ...
| ISeq (i0 , i1) -> flat i0 @ flat i1

5.4 Linearizing Instructions

In the previous section, we used ISeq to combine two instructions. As such, an instruction
had a tree structure. We can turn it into a linear list by flattening the tree into an OCaml
list. With this transformation, i in VFun becomes i list (or equivalently, c) and ISeq is
removed from i.

Although the transformation is intuitively clear, to show its correctness, we need to prove
that the instructions form a monoid. Namely, the grouping of instructions does not matter
as long as the order of instructions is preserved. We briefly sketch the proof. We first define
a flattening function (Listing 10) that turns i into a list of i’s without ISeq. We can define
similar functions (flatV, flatC, etc.) that flatten all the instructions appearing in given
data (a value, a continuation, etc., respectively). We then prove the following equivalences:

flat (f9 e xs) = f10 e xs, stating that the list of instructions generated by the new
compiler is the same as flattening the instruction generated by the old compiler, and

flatV (run_i9 i c s t m) = run_c10 (flat i @ flatC c) (flatS s) (flatT t)
(flatM m), stating that running i under c in the old virtual machine yields the same
result as running the flattened instructions of i and c in the new virtual machine (or
both do not terminate).

The former is proved by induction on the structure of e and the latter on the number of steps
the old virtual machine takes. One has to be careful in the case when i is ISeq. Although
the old virtual machine takes a step to execute it, there is no corresponding execution step
in the new virtual machine, since ISeq is already flattened. Therefore, the termination
behavior of the two virtual machines is different when the instruction list contains infinitely
many ISeq’s: the former continues indefinitely executing ISeq’s while the latter terminates
since all the ISeq’s are already flattened and removed. This does not happen, since all the
instructions are finite.

FSCD 2021



16:14 Virtual Machine for Four Delimited-Control Operators

Listing 11 Interpreter with linearized trails.
type v = VFun of c * v list | VContS of t | VContC of t

| VEnv of v list | VK of c
and i = IAccess of int | IPush_closure of c | IReturn

| IPush_env | IPop_env | ICall
| IShift of c | IControl of c | IShift0 of c | IControl0 of c
| IReset of c

and c = i list
and s = v list
and t = (c * s) list
type m = t list

Figure 2 Virtual machine.

c ⇒ ⟨c, [VEnv([])], [], []⟩
⟨[], v :: [], [], []⟩ ⇒ v

⟨[], v :: [], [], ((c, s) :: t) :: m⟩ ⇒ ⟨c, v :: s, t, m⟩
⟨[], v :: [], (c, s) :: t, m⟩ ⇒ ⟨c, v :: s, t, m⟩

⟨IAccess(n) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c, (List.nth vs n) :: s, t, m⟩
⟨IPushClosure(c′) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c, VFun(c′, vs) :: s, t, m⟩

⟨IReturn :: _, v :: VK(c) :: s, t, m⟩ ⇒ ⟨c, v :: s, t, m⟩
⟨IPushEnv :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c, VEnv(vs) :: VEnv(vs) :: s, t, m⟩

⟨IPopEnv :: c, v :: VEnv(vs) :: s, t, m⟩ ⇒ ⟨c, VEnv(vs) :: v :: s, t, m⟩
⟨ICall :: c, v :: VFun(c′, vs) :: s, t, m⟩ ⇒ ⟨c′, VEnv(v :: vs) :: VK(c) :: s, t, m⟩

⟨ICall :: c, v :: VContS((c′, s′) :: t′) :: s, t, m⟩ ⇒ ⟨c′, v :: s′, t′, ((c, s) :: t) :: m⟩
⟨ICall :: c, v :: VContC ((c′, s′) :: t′) :: s, t, m⟩ ⇒ ⟨c′, v :: s′, t′ @ (c, s) :: t, m⟩

⟨IShift(c′) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c′, VEnv(VContS((c, s) :: t) :: vs) :: [], [], m⟩
⟨IControl(c′) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c′, VEnv(VContC ((c, s) :: t) :: vs) :: [], [], m⟩

⟨IShift0(c′) :: c, VEnv(vs) :: s, t, ((c0, s0) :: t0) :: m0⟩ ⇒ ⟨c′ @ c0, VEnv(VContS((c, s) :: t) :: vs) :: s0, t0, m0⟩
⟨IControl0(c′) :: c, VEnv(vs) :: s, t, ((c0, s0) :: t0) :: m0⟩ ⇒ ⟨c′ @ c0, VEnv(VContC ((c, s) :: t) :: vs) :: s0, t0, m0⟩

⟨IReset(c′) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c′, VEnv(vs) :: [], [], ((c, s) :: t) :: m⟩

5.5 Linearizing Trails
Finally, we transform the type t of trails, which had a tree structure (Listing 9), into

a linear list. By regarding TNil as an empty list, Hold as a singleton list consisting of c *
s, and Append as a list append, we can represent t as a list of c * s. The resulting type
definitions are shown in Listing 11. Now that t becomes (c * s) list, we change the type
of VContS and VContC from c * s * t to t by piling up the c * s pair onto t. Similarly, m
can be represented as t list.

To establish the correctness of this transformation, we need to show that the new virtual
machine behaves the same as before:

flatV (run_c10 c s t m) = run_c11 c (flatS s) (flatT t) (flatM m)

where flat functions are defined similarly to the ones in the previous section to flatten the
type of trails. The above equivalence is shown by induction on the number of execution steps
the old virtual machine takes.

6 Virtual Machine

Figure 2 shows the state transition rules for the virtual machine obtained from the interpreter
in the previous section. The main state consists of a tuple (c, s, t, m) of four elements: a
continuation, a stack, a trail, and a metacontinuation. We show an example how a program
is compiled to a list of instructions and executed on the virtual machine in the appendix.

The virtual machine succinctly models the low-level behavior of control operators. Just as
in the abstract machine, when one of the control operators is used, the current continuation
(or a pointer to an instruction) c, stack s, and trail t are captured and put into a stack. Then,
the body of the control operator is executed. For IShift and IControl, the current continuation
and trail are cleared, whereas for IShift0 and IControl0, the ones in the metacontinuation
are used. The reset operator pushes the current c, s, and t on the metacontinuation m, and
initializes them.

When a continuation captured by IShift or IShift0 is invoked, the current c, s, and t

are pushed onto m and the captured state is reinstated. When a continuation captured by
IControl or IControl0 is invoked, on the other hand, the current c and s are added to t to
which the captured trail t′ is appended.

Although we maintain s, t, and m separately in the virtual machine, we can represent
them as a single stack. Remember that s is a list of values. Since t is a list of pairs of c and
s, it has the form:

[(c, [v; . . . ; v]); . . . ; (c, [v; . . . ; v])]

Thus, if we represent c as a single value (e.g., using VK) pointing to the first instruction
designated by c, and if we maintain the positions of c in t using pointers, we can represent t

as a list of values. Furthermore, since m is a list of trails (a list of lists of pairs of c and s), it
can be represented as a list of values, too, if we maintain pointers to each element of m.

If we represent s, t, and m as a single stack, we notice that we can sometimes avoid
copying s and t. When c, s, and t are pushed to m in the rules for IReset and the VContS
and VContC cases of ICall, the ordering of s, t, and m does not change. Thus, we can simply
rearrange the pointers to the head of a stack, trail, and metacontinuation appropriately,
without copying s and t. Similarly for s0, t0, and m0 in the rules for IShift0 and IControl0.
When do we have to copy s and t? It is when we use control operators or apply captured
continuations. The s and t must be copied, in the former case to be stored in VContS or
VContC , and in the latter case to use what was stored.

Finally, in the rules for IShift0 and IControl0, the body instructions c′ of the control oper-
ators and the instructions c0 in the metacontinuation are concatenated. This concatenation
reflects the fact that the body of IShift0 and IControl0 has access to the context outside the
current enclosing reset. (In the abstract machine, the concatenation was realized by execut-
ing the body under the continuation stored in the metacontinuation.) Implementation-wise,
this suggests that we need to keep track of a list of pointers to these continuations, which is
an interesting observation that has not been observed before.

7 Related Work

We are not aware of any work that derives a virtual machine for the four delimited-control
operators other than shift/reset. Deriving a virtual machine for other language constructs
includes Ager, Biernacki, Danvy, and Midtgaard’s work [1] for λ-calculus (of various flavors)
and Igarashi and Iwaki’s work [18] for a staged language.

As for an abstract machine, Biernacki, Danvy, and Millikin [5] present abstract machines
for the four delimited-control operators as definitional and derive a CPS interpreter for
control/prompt. Shan [28] derives an abstract machine for control/prompt from the CPS
interpreter for control/prompt. In both work, the derivation is done for control/prompt
only. Their abstract machines are similar to ours but do not maintain a stack explicitly.

Dyvbig, Peyton Jones, and Sabry [11] show an abstract machine for primitive control
operators that can implement four delimited-control operators with named prompts. Since
they use their own primitive control operators, their CPS interpreter is quite different from

ours. They do not use trails and represent concatenation of contexts using a metacontinuation,
which is a list of continuations. Based on this abstract machine, Kiselyov [21] implements
the control operators in OCaml by emulating the behavior of the abstract machine using
OCaml’s exception handling mechanism.

Hillerström, Lindley, and Atkey [17] show CPS translations and abstract machine se-
mantics for algebraic effects and handlers. It would be interesting to see if the program
transformation approach can be used in this setting, too.

8 Conclusion

In this paper, we have derived a compiler and a virtual machine for the four delimited-control
operators from the definitional interpreter. The resulting virtual machine suggests a low-level
implementation method for delimited continuations.

Although we focused on the behavior of the delimited-control operators, we also want
to consider their type systems. We are currently trying to build a type system for the four
delimited-control operators (the one for control/prompt is in [6]). Once we obtain a type
system, we plan to implement the four delimited-control operators in OchaCaml [22] based
on the virtual machine developed in this paper. That would form a solid foundation on
which a different implementation of algebraic effects and handlers can be considered.

A Example Execution

In this appendix, we show an example how the compiler and the virtual machine work. We
use the control term in Section 2: 1+ ⟨(Fc. 2 × c 3) + Fc′. 4⟩. It is straightforward to support
numbers and binary operators; see the supplementary material. State transition rules for the
new instructions are summarized in Figure 3.

The list of instructions output by the compiler is:

[IPushEnv1; INum(1); IPopEnv1; IReset(c1); IOp1(+)]

where

c1 = [IPushEnv2; IControl1(c2); IPopEnv2; IControl2(c3); IOp2(+)]
c2 = [IPushEnv3; INum(2); IPopEnv3; IPushEnv4; IAccess(0); IPopEnv4;

INum(3); ICall; IOp3(∗)]
c3 = [INum(4)]

We use subscripts to disambiguate instructions that appear more than once.
The list of instruction is executed as in Figure 4. We can observe that the trails 3 + []

(i.e., ([IOp2(+)], [VNum(3)])) and 2 × [] (i.e., ([IOp3(∗)], [VNum(2)])) are concatenated at
the second invocation of IControl and are captured in vc2.

⟨INum(n) :: c, VEnv(vs) :: s, t, m⟩ ⇒ ⟨c, VNum(n) :: s, t, m⟩
⟨IOp(+) :: c, VNum(n0) :: VNum(n1) :: s, t, m⟩ ⇒ ⟨c, VNum(n0 + n1) :: s, t, m⟩
⟨IOp(∗) :: c, VNum(n0) :: VNum(n1) :: s, t, m⟩ ⇒ ⟨c, VNum(n0 ∗ n1) :: s, t, m⟩

Figure 3 State transition rules for INum and IOp.

